Decision Modeling

Chapter One

Solutions to the Problems for Student Completion

David M. Tulett

Last modified on September 17, 2021.

1 Spreadsheet Formula Exercises

(a) The tax credit is computed using:

```
=IF(C2<=B2,C2*A2,B2*A2+(C2-B2)*A5).
```

When solved we obtain:

	A	B	C
1	Lowest Tax Rate		Donations
2	21.50%	$200.00	$3,700.00
3			
4	Highest Tax Rate		Tax Credit
5	48.50%		$1,740.50

(b) In column E we will create a formula for the first student (in row 3) and then copy this formula for all the others. Hence, both references to cells G2 and G3 will require that we freeze the row numbers by putting a dollar sign in front of the row number. One way to create the formula is to take the maximum of two expressions; one which includes both midterms, and one that keeps only the better of the two midterm marks, adding the single midterm weight to that of the final exam. In this latter case, the weight of the final exam is now `G$2+G$3`, or equivalently, `1-G$2`.

The MAX expression gives a decimal answer; it is therefore enclosed within the ROUND function with the number of digits set to 0.

```
=ROUND(MAX(G$2*(B3+C3)+G$3*D3,G$2*MAX(B3,C3)
+(1-G$2)*D3),0)
```

Another way to do this is to use an IF statement. If the final exam mark is lower than either of the two test marks, then we count both midterms. Otherwise, we count only the better of the two midterms, and add the single midterm weight to that of the final exam. As before, we need to round this decimal answer to the nearest integer.

```
=ROUND(IF(D3<=MIN(B3,C3),G$2*(B3+C3)+G$3*D3),
G$2*MAX(B3,C3)+(1-G$2)*D3),0)
```

Another way to do this is:

```
=ROUND(IF(MIN(B3:C3)<=D3,MAX(B3:C3)*G$2+D3*(G$2+G$3),
SUM(B3:C3)*G$2+D3*G$3),0)
```

However we do it, the solution is:

	A	B	C	D	E	F	G
1	**Student Names**	**Test 1**	**Test 2**	**Final Exam**	**Final Mark**		
2							20%
3	Bartlett, Joanna	37	65	73	71		60%
4	Chan, Mia	82	86	81	82		
5	Duval, Pierre	72	56	64	66		

(c) The length of the tunnel is the length of the hypotenuse of a right-angled triangle, where the lengths of the other two sides are the distances along the east-west and north-south axes. Using the theorem of Pythagoras, this is `=SQRT((103-345)^2+(296-237)^2)`.

Alternatively, we can put the data into cells, computing the distance using:

```
=SQRT((B2-B3)^2+(C2-C3)^2)
```

	A	B	C
1		**a**	**b**
2	**Point 1**	103	296
3	**Point 2**	345	237
4			
5	**Distance**	249.0883	

2 Phone Plans

(a) In the text of Chapter 1 we are told that in terms of the number of local minutes a, and the number of long-distance minutes b, the cost if Plan 1 is followed is:

$$35 + 0.5\max\{a - 200, 0\} + 0.5b$$

The cost to use phone Plan 4 is:

$$52 + 0.5\max\{a + b - 1000, 0\}$$

Hence Plan 1 is cheaper than Plan 4 for all values of a and b such that:

$$35 + 0.5\max\{a - 200, 0\} + 0.5b \leq 52 + 0.5\max\{a + b - 1000, 0\}$$

We can simplify this inequality by subtracting 35 from both sides to obtain:

$$0.5\max\{a - 200, 0\} + 0.5b \leq 17 + 0.5\max\{a + b - 1000, 0\}$$

We now multiply both sides by 2 to obtain:

$$\max\{a - 200, 0\} + b \leq 34 + \max\{a + b - 1000, 0\}$$

Case 1 If $a \leq 200$ and $a + b \leq 1000$, then the expression simplifies to:

$$b \leq 34$$

Hence Plan 1 is cheaper than Plan 4 if $a \leq 200$ and $b \leq 34$. Since both a and b cannot be negative, Case 1 forms a rectangle 200 units wide and 34 units high going from (a, b) at (0,0), to (0,34), to (200,34), to (200,0), and then back to (0,0).

Case 2 If $a \geq 200$ and $a + b \leq 1000$, then we have:

$$a - 200 + b \leq 34$$

Hence $a + b \leq 234$. Case 2 is bounded by $a \geq 200$, $b \geq 0$, and $a + b \leq 234$. These three bounds form an isosceles right-angled triangle going from (a, b) at (200,0), to (200,34), to (234,0), and then back to (200,0).

Case 3 If $a \leq 200$ and $a+b \geq 1000$, then the expression simplifies to:

$$b \leq 34 + a + b - 1000$$

Canceling the b from both sides, we obtain $a \geq 966$. This result is a contradiction, since by assumption $a \leq 200$, hence Case 3 cannot happen.

Case 4 If $a \geq 200$ and $a+b \geq 1000$, then the expression simplifies to:

$$a - 200 + b \leq 34 + a + b - 1000$$

We cancel $a+b$ from both sides of the inequality, leaving us with $-200 \leq -966$, which is false. Hence, Case 4 cannot happen.

Overall, only Cases 1 and 2 have validity. The entire region in which Plan 1 is cheaper than Plan 4 is the rectangle found in Case 1 plus the triangle found in Case 2. These two regions together form a quadrilateral bounded by (a,b) at (0,0), (0,34), (200,34) and (234,0).

(b) Plotting this quadrilateral we obtain:



3 Cargo Plane Loading Problem

Here we present both a pdf of the spreadsheet in numerical form, and the formulas used to calculate these numbers. We see in part (a) that there are 26 feasible solutions. We also see in part (b) that at most five of these 26 solutions could potentially be optimal, and these are evaluated in part (c).

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Cargo Plane Loading Problem			X = number of Type 1 boxes			Y = number of Type 2 boxes						
2													
3		Type 1	Type 1	Type 2	Type 2		Plane Volume Capacity (m^3)			15			
4		Volume	Mass	Volume	Mass		Plane Mass Capacity (kg)			3600			
5		2.9	470	1.8	530								
6													
7	X	Remaining	Remaining	Max. Y	Max. Y	Overall		Part (b)			Part c)	ii	iii
8		Volume	Mass	(Volume)	(Mass)	Max. Y		X	Y		400	600	300
9											400	250	750
10	0	15	3600	8.333333	6.792453	6		0	6		2400	1500	4500
11	1	12.1	3130	6.722222	5.90566	5							
12	2	9.2	2660	5.111111	5.018868	5		2	5		2800	2450	4350
13	3	6.3	2190	3.5	4.132075	3		3	3		2400	2550	3150
14	4	3.4	1720	1.888889	3.245283	1		4	1		2000	2650	1950
15	5	0.5	1250	0.277778	2.358491	0		5	0		2000	3000	1500
16	6	-2.4	780	-1.33333	1.471698	-2							
17								Note: (1,5) is			X = 2	X = 5	X = 0
18								trivially sub-optimal.			Y = 5	Y = 0	Y = 6
19													
20	Recommendation:		When both types of boxes are worth $400 each, carry 2 Type 1 boxes and 5 Type 2 boxes.										
21			The revenue will be $2800.										
22													
23			When a Type 1 box is worth $600, but a Type 2 only $250, carry 5 Type 1 boxes and no Type 2 boxes.										
24			The revenue will be $3000.										
25													
26			When a Type 1 box is worth only $300, but a Type 2 box is worth $750, carry no Type 1 boxes and 6 Type 2 boxes.										
27			The revenue will be $4500.										

	A	B	C	D	E	F
1	Cargo Plane Loading I			X = number of Type 1		
2						
3		Type 1	Type 1	Type 2	Type 2	
4		Volume	Mass	Volume	Mass	
5		2.9	470	1.8	530	
6						
7	X	Remaining	Remaining	Max. Y	Max. Y	Overall
8		Volume	Mass	(Volume)	(Mass)	Max. Y
9						
10	0	=$J$3-$B$5*A10	=$J$4-$C$5*A10	=B10/$D$5	=C10/$E$5	=INT(MIN(D10,E10))
11	1	=$J$3-$B$5*A11	=$J$4-$C$5*A11	=B11/$D$5	=C11/$E$5	=INT(MIN(D11,E11))
12	2	=$J$3-$B$5*A12	=$J$4-$C$5*A12	=B12/$D$5	=C12/$E$5	=INT(MIN(D12,E12))
13	3	=$J$3-$B$5*A13	=$J$4-$C$5*A13	=B13/$D$5	=C13/$E$5	=INT(MIN(D13,E13))
14	4	=$J$3-$B$5*A14	=$J$4-$C$5*A14	=B14/$D$5	=C14/$E$5	=INT(MIN(D14,E14))
15	5	=$J$3-$B$5*A15	=$J$4-$C$5*A15	=B15/$D$5	=C15/$E$5	=INT(MIN(D15,E15))
16	6	=$J$3-$B$5*A16	=$J$4-$C$5*A16	=B16/$D$5	=C16/$E$5	=INT(MIN(D16,E16))

	K	L	M
7	Part c)	ii	iii
8	400	600	300
9	400	250	750
10	=$H10*K$8+$I10*K$9	=$H10*L$8+$I10*L$9	=$H10*M$8+$I10*M$9
11			
12	=$H12*K$8+$I12*K$9	=$H12*L$8+$I12*L$9	=$H12*M$8+$I12*M$9
13	=$H13*K$8+$I13*K$9	=$H13*L$8+$I13*L$9	=$H13*M$8+$I13*M$9
14	=$H14*K$8+$I14*K$9	=$H14*L$8+$I14*L$9	=$H14*M$8+$I14*M$9
15	=$H15*K$8+$I15*K$9	=$H15*L$8+$I15*L$9	=$H15*M$8+$I15*M$9
16			
17	X = 2	X = 5	X = 0
18	Y = 5	Y = 0	Y = 6